

Nominalphrasengrammatik

Nominalphrasen: Ausdrücke, die als Objekte auftreten und durch Personalpronomen ersetzbar, aber keine Sätze sind

Kategorie: *NP*

Hilfskategorien: *Det, Art, Qnt, Poss, Pron, N, N', PN, MN, CN, AP*

Grammatikregeln:

- ▶ atomare *NP*: Personalpronomen, Eigennamen, artikellose Klassennomen (Plural), artikellose Massennomen (Singular)
- ▶ zusammengesetzte *NP*: $Det + AP^* + (N + NP^{gen} + \text{Objekt}) + S_{rel}$
 1. Objekte von Nomen sind optional, folgen auf $N + NP^{gen}$
 2. Attribute wie *AP*-Folgen und S_{rel} (oder *PP*) sind optional
 3. determinierende Elemente sind Artikel, Quantoren, Possessiv- und Personalpronomen, NP^{gen} , manche Kombinationen davon
 4. *Det, AP, N* kongruieren im Numerus und Kasus
 5. *N* regiert *Det, AP* im Genus, *Det* regiert *AP* in der Deklination

Unklar:

1. Sind AP -Attribut und S_{rel} - oder PP -Attribut gleichrangig?

das (grobe Benehmen), das mich stört

↳ (das grobe Benehmen) stört mich

das grobe (Benehmen, das mich stört)

↳ (das Benehmen, das mich stört), ist grob

2. Stehen NP^{gen} -Attribute stets enger am N als Objekte, PP - und S_{rel} -Attribute?

das Auto des Chefs auf dem Hof

= das Auto, das dem Chef gehört und das auf dem Hof steht

≠ das Auto auf dem Hof des Chefs

das Auto des Mannes aus der Führungsetage

= das Auto, das dem Mann aus der Führungsetage gehört

≠ das Auto aus der Führungsetage des Mannes

NP-Grammatik ohne Merkmale

Kategorie	Konstruktion	Beispiele
<i>NP</i>	= <i>Pron</i>	er
	<i>PN</i>	Maria
	<i>CN</i>	Menschen
	<i>MN</i>	Wein
	<i>N'</i>	Kinder, die noch nicht laufen können,
	<i>Det N'</i>	kein Schwein im Stall
<i>N'</i>	= $N_{\square} NP^{gen}$	Schatten (des Körpers (des Kutschers))
	$N (S_{rel} \mid PP)$	Hund (, der) hinter dem Ofen (liegt,)
	$N_{[P]} NP^{gen} PP^P$	Angst _[vor] des Tormanns vorm Elfmeter
	$N_{[gen]} NP^{gen}$	Ermordung _[gen] des Cäsar
	$N_{[dass]}, \text{da\ss } S^{vl},$	These, da\ss die Erde eine Kugel ist,
	$N_{[dass]}, S^{vz,konj},$	Nachricht, ein Prophet komme,
	$N_{[infzu]}, S^{infzu},$	Entschlu\ss, nicht mehr zu rauchen,
	<i>AP N'</i>	sehr gro\ss (Angst vor dem Untergang)

Kategorie	Konstruktion	Beispiele
<i>Det</i>	= <i>Art</i>	ein; der; dieser; jener
	<i>Qnt</i>	jeder; mancher; kein; viele
	<i>Poss</i>	mein; dein; sein; ihr; unser; euer; ihr
	<i>NP^{gen}</i>	des Bäckers
	<i>Pron</i>	ich
N_{\square}	= <i>PN</i>	Maria;
	<i>CN</i>	Hunde
	<i>MN</i>	Wein
	<i>NA</i>	Eitelkeit; Schönheit; Ängstlichkeit
$N_{[Arg]}$	= $NV_{[Arg]}$	Versuch _[infzu] ; Hoffnung _[auf] Tragen _[gen] ; Behauptung _[dass]
	$NA_{[P]}$	Eifersucht _[auf] ,

Das Lexikon muß klären, wie die Komplemente gemeint sind:

entschließen: $V_{[nom, refl(akk), infzu]}$ \mapsto Entschluß: $NV_{[?gen, infzu]}$

ermorden: $V_{[nom, akk]}$ \mapsto Mord: $NV_{[?gen, an]}$, Ermordung: $NV_{[?durch, gen]}$

NP-Grammatik mit Merkmalen

Kategorie und Merkmale: $NP_{Person, Genus, Numerus}^{Kasus}$

- ▶ Person, Numerus, Kasus für Kongruenz mit dem Verb nötig
- ▶ Nominalphrasen sind i.a. nur im Kasus variierbar (Eigennamen)
- ▶ nur (in)definite NPs, keine interrogativen oder relativierenden

Hilfskategorien und ihre Merkmale:

- ▶ $Det_{Person, Numerus, Deklinationstyp}^{Genus, Kasus}$
- ▶ $Art_{Deklinationstyp, Numerus}^{Genus, Kasus}$ und $Qnt_{Deklinationstyp, Numerus}^{Genus, Kasus}$
- ▶ $Poss_{Person', Genus', Numerus'}^{Genus, Numerus, Kasus}$ und $Pron_{Person, Genus, Numerus, Kasus}$
- ▶ $N_{Komplementrahmen, Genus}^{Numerus, Kasus}$ und $A_{Komplementrahmen}^{Deklinationstyp, Genus, Numerus, Kasus}$
- ▶ $N'_{Genus}^{Deklinationstyp, Numerus, Kasus}$
- ▶ $AP_{Deklinationstyp, Komparation, Genus, Numerus, Kasus}$

Wir brauchen für das Deutsche in der Nominalphrase nur 8 statt $3 \times 3 \times 2 = 18$ mögliche Kombinationen von Person, Genus, Numerus:

Pers	1	1	2	2	3	3	3	3
Gen	-	-	-	-	mask	fem	neut	-
Num	sg	pl	sg	pl	sg	sg	sg	pl

da die konkreten Formen der beteiligten Wörter nicht mehr Unterschiede zeigen. (Das ist in anderen Sprachen i.a. anders.)

Man hat nun zwei Möglichkeiten:

1. Man verwendet bei den Kategorien alle 18 Kombinationen der normalen Werte $\{1,2,3\} \times \{mask, fem, neut\} \times \{sg, pl\}$.
Dann ist die Merkmalvererbung in den Syntaxregeln einfach, z.B. mit Variablen für die Werte der Dimensionen

$$NP_{Pers, Gen, Num}^{Kas} \rightarrow Pron^{Pers, Gen, Num, Kas},$$

aber im Lexikon hat man künstliche Lesartenvielfalt, z.B.
 $Pron^{2, mask, sg, nom} \rightarrow du$ und $Pron^{2, fem, sg, nom} \rightarrow du$.

2. Man benutzt nicht alle 18 Kombinationen, aber manchmal den künstlichen Genuswert – (kein bestimmbares Genus). Dann hat man im Lexikon weniger Einträge, wie

$$Pron^{2,-,sg,nom} \rightarrow du, \quad (\text{kein Genus}),$$

aber in den Grammatikregeln muß man den Wert – gesondert behandeln. Wenn man z.B. das Genus als eine Formdimension der Determinatoren $Det_{Pers,Num,Dekl}^{Gen,Kas}$ betrachtet und die Personalpronomen auch als Determinatoren verwenden will

$$Det_{Pers,Num,st}^{Gen,Kas} \rightarrow Pron^{Pers,Gen,Num,Kas}$$

(z.B. *du braves Kind*), dann muß man bei der NP-Konstruktion den Fall $Gen = - \neq Gen \in \{mask, fem, neut\}$ extra behandeln:

$$NP_{Pers,Gen,Num}^{Kas} \rightarrow Det_{Pers,Num,Dekl}^{Gen,Kas} AP^{Gen,Num,Kas,Dekl} N_{Gen}^{Num,Kas}$$

Implementierung als DCG

$\langle np.folien.pl \rangle \equiv$

```
:- ['DCGparser/parser.pl'].
```

```
% Prädikate, die über mehrere Dateien definiert sind:
```

```
:- multifile n/5, cn/5, nv/5, a/5, art/5, pron/5,  
    poss/5, num/5, det/5, np/5, s/5, v/5, ap/5,  
    praeposition/2. % 5-stellig: cat(Art,Form,-Baum,+I,J)
```

```
% Analyse der Eingabe als:
```

```
startsymbol(np([_Pers,_Gen,_Num],[_Kas])).
```

```
genus(X):- member(X,[mask,fem,neut,-]). % Merkmalwerte
```

```
numerus(X):- member(X,[sg,pl]).
```

```
kasus(X) :- member(X,[nom,gen,dat,akk]).
```

```
person(X):- member(X,[1,2,3]).
```

```
deklinationsstyp(X):- member(X,[st,sw,ge]).
```


Wir brauchen ein Beispiellexikon (s.u.) und ein paar Pseudodefinitionen, die später in der Satzgrammatik überschrieben werden:

$\langle np.folien.pl \rangle + \equiv$

:- [lexikon_np]. % Lexikalische Regeln

% Pseudodefinitionen mit ad-hoc Satzarten

s([rel],[ind]) --> [rels] . % Relativsätze

s([def],[vz,konj]) --> [s]. % definite Sätze im Konj.

s([dass],[v1,ind]) --> [dass],[s]. % daß-Sätze

s([qu],[v1,ind]) --> [s]. % Interrogativsätze

s([ob],[v1,ind]) --> [ob],[s]. % ob-Sätze

s([infzu],[]) --> [infzu]. % Verbalphrasen

$\langle npgram.pl \rangle$ % NP-Regeln der folgenden Folien

(definite) Nominalphrasen

Wir benutzen die 2. Möglichkeit, d.h. verwenden künstliche Werte – und behandeln die separat in den Regeln. Z.B. sagt die erste Regel, daß das Genus unbestimmt (-) ist, *oder* Person und Numerus die Werte 3 und sg haben. Im Lexikon muß das Genus – *genau* bei 1./2.Person Singular und im Plural vergeben werden.

Kongruenz und Rektion werden durch Variablengleichheiten gezeigt.

Kategorie	Konstruktion
$NP_{Pers, Gen, Num}^{Kas}$	$=$ $Pron^{Pers, Gen, Num, Kas}$ (Gen=-; Pers=3, Num=sg) PN_{Gen}^{Kas} (Pers=3, Num=sg) $N'_{Gen}^{Adekl, Num, Kas}$ (Pers=3, (Adekl=st ; Adekl=-, Kas≠gen)) $Det_{Pers, Num, Adekl}^{Gen, Kas}$ $N'_{Gen}^{Adekl, Num, Kas}$

Implementierung als DCG

```
<npgram.pl>≡  
np([Pers,Gen,Num],[Kas]) -->  
  ( pron([], [Pers,Gen,Num,Kas])           % er, sie, es  
    ; pn([Gen],[Kas]), { Pers=3,Num=sg } % Hans, Maria  
    ; n1([Gen],[Dekl,Num,Kas]),           % alte Menschen  
      { Pers=3,(Num=pl;Num=sg),           % ohne Adj nicht  
        (Dekl=st ; (Dekl=(-), Kas\=gen)) } % im pl.gen  
    ; det([Pers,Num,Dekl],[GenD,Kas]), % der alte Hund  
      { GenD=(-) ; GenD=Gen },           % ein alter Hund  
      n1([Gen],[Dekl1,Num,Kas]),        % ich alter Hund  
      { Dekl1=(-) ; Dekl1=Dekl } % ggf. kein Adj in N1  
  ).
```

Determinatoren (s.u. Folie 16) (und Adjektive) haben im Plural und in erster und zweiter Person kein echtes Genus, sondern einen Wert -, damit ihr Paradigma klein bleibt.

Einfache und attribuierte Nomina mit Komplementen

Kategorie	Konstruktion
$N'_{Gen}{}^{Dekl, Num, Kas}$	$= N_{[], Gen}^{Num, Kas} (NP_{3, Gen', Num'}^{gen})$ $N_{[P, Kas'], Gen}^{Num, Kas} (NP_{3, Gen', Num'}^{gen}) P_{Kas'} NP_{Pers, Gen2, Num'}^{Kas'}$ $N_{[gen], Gen}^{Num, Kas} NP^{gen}$ $N_{[dass], Gen}^{Num, Kas}, \text{ daß } S^{vl},$ $N_{[dass], Gen}^{Num, Kas}, S^{vz, konj},$ $N_{[infzu], Gen}^{Num, Kas}, S^{infzu},$ $AP^{Dekl, pos, Gen, Num, Kas} N'_{Gen}{}^{Dekl, Num, Kas}$ $N_{Kompl, Gen}^{Num, Kas} (S_{rel} P_{Kas'} NP_{Pers', Gen', Num'}^{Kas'})$

Implementierung als DCG

$\langle npgram.pl \rangle + \equiv$

```
% N mit Komplement und Attribut: n1([Gen],[Adekl,Num,Kas])
```


```
n1([Gen],[Dekl,Num,Kas]) --> % Nomen+Komplement
    { Dekl = (-) },           % keine Adjektive vorhanden
    n([Kompl,Gen],[Num,Kas]), % ggf. Kompl = []
    ( [] ; np([3,-,-],[gen]) ), % opt. gen.poss/subj
    ( [] % Komplement optional: Hoffnung (auf NP)
    ; { Kompl=[gen] }, []      % gen-Objekt statt gen.subj
    ; { Kompl=[Praep,KasP],   % Angst vor den Hunden
      praeposition(Praep,KasP) },
      [Praep], np([_Pers,_Gen,_Num],[KasP]))
```

Implementierung als DCG

Nomina mit Objektsätzen oder Relativsätzen: (vorläufige s-Merkm.)

`<npgram.pl>+≡`

```
; { Kompl=[dass] },
    ( s([dass],[v1,Mod]) % These, daß sie rund ist|sei
      ; s([def],[vz,konj]) % These, sie *ist|sei rund
    )
; { Kompl=[ob] },      % Untersuchung, ob sie lebt
    s([ob],[v1,ind])
; { Kompl=[qu] },      % Frage, wo Maria lebt
    s([qu],[v1,Mod])
; { Kompl=[infzu] },   % Absicht, zu rauchen
    s([infzu],[ ])
),
( [ ] % optionaler Relativsatz: Menschen,
  ; s([rel(Gen,Num)],[Mod]) ).      % an die wir denken
```

Bei Sätzen wird nach Verbstellung (ve, vz, v1) unterschieden. 

Attributierung durch eine Adjektivphrase im Positiv:

$\langle npgram.pl \rangle + \equiv$

```
n1([Gen],[Dekl,Num,Kas]) -->
    { Komp = pos }, % TODO: Komp=komp, sup
    ap([], [Komp,Dekl,GenD,Num,Kas]), % alte Hunde, die ..
    { GenD=(-) ; GenD=Gen },
    n1([Gen],[Dekl1,Num,Kas]),
    { Dekl1 = Dekl ; Dekl1 = (-) }.
```

```
praeposition(in,akk). praeposition(an,akk).
praeposition(auf,akk). praeposition(um,akk).
praeposition(mit,dat). praeposition(vor,dat).
praeposition(von,dat).
```

Determinierende Ausdrücke

Wir betrachten *Det* nicht als eine Wortart, sondern als eine Art von *Ausdrücken*, die im Kasus und (bis auf NP^{gen}) im Genus variieren:

Kategorie	Konstruktion
$Det_{Pers, Num, Dekl}^{Gen, Kas}$	= $Art_{Dekl, Num}^{Gen, Kas}$ (Pers=3)
	$Qnt_{Dekl, Num}^{Gen, Kas}$ (Pers=3)
	$Poss_{Pers', Gen', Num'}^{Gen, Num, Kas}$ (Pers=3, Dekl=ge)
	$NP_{3, Gen', Num'}^{gen}$ (Pers=3, Dekl=ge)
	$Pron^{Pers, Gen, Num, Kas}$ (Pers \neq 3, Kas \neq gen, Dekl=st)

Implementierung als DCG

$\langle npgram.pl \rangle + \equiv$

```
det([Pers,Num,Dekl],[Gen,Kas]) -->
  ( ( art([Dekl,Num],[Gen,Kas])           % der, ein
    ; qnt([Dekl,Num],[Gen,Kas])           % kein, jeder
    ; poss([_,_,_],[Gen,Num,Kas]),{Dekl=ge} % mein, unser
    ), {Pers=3}
  ; pron([], [Pers,Gen,Num,Kas]),         % in manchen
    { (Pers=1 ; Pers=2)                   % Formen:
      , (Kas=nom ; Kas=dat ; Kas=akk)
      , Dekl=st }                         % ich armer Student
  ).
```

$\langle \text{Linksrekursion, führt beim Top-Down-Parsen zur Endlosschleife} \rangle \equiv$

```
np([Pers,Gen,Num],[Kas]) -->
  np([3,_Gen,_Num],[gen]),{Dekl=ge} % des Bäckers Brote
  n1([Gen],[Dekl,Num,Kas]).
```

Nomina ohne bzw. mit Komplementen

Eigennamen PN , zählbare Nomen CN , Massennomen MN gelten als Nomen mit leeren Komplementrahmen; nominalisierte Adjektive NA und Verben NV als solche mit Komplementrahmen:

Kategorie	Konstruktion	Beispiel
$N_{\square, Gen}^{Num, Kas}$	= PN_{Gen}^{Kas} (Num=sg)	Peter, Maria
	$CN_{Gen}^{Num, Kas}$	Hund, Mensch
	MN_{Gen}^{Kas} (Num=sg)	Wasser, Ruhe
$N_{Kompl, Gen}^{Num, Kas}$	= $NA_{Kompl, Gen}^{Num, Kas}$	Besorgtheit um
	$NV_{Kompl, Gen}^{Num, Kas}$	Hoffnung auf
	$N_{[P], Gen}^{Num, Kas}$	Angst vor, Neid auf

Implementierung als DCG

Nomina mit und ohne Komplemente:

$\langle npgram.pl \rangle + \equiv$

```
% Zusammenfassung von Nomenarten: n([Kompl,Gen],[Num,Kas])
```

```
n([],Gen,[Num,Kas]) -->
```

```
  ( pn([Gen],[Kas]), {Num=sg} % Peter, Maria
    ; cn([Gen],[Num,Kas])     % Hund, Menschen
    ; mn([Gen],[Num,Kas])     % Wein(e), Wasser, Ruhe
  ).
```

```
n([Kompl,Gen],[Num,Kas]) -->
```

```
  ( na([Kompl,Gen],[Num,Kas]) % Gute, Neid auf
    ; nv([Kompl,Gen],[Num,Kas]) % Hoffnung auf
  ).
```

Im Lexikon kann man auch Nomina mit Komplementen erlauben, die weder nominalisierte Adjektive noch nominalisierte Verben sind.

Lexikoneinträge

⟨Ein paar Lexikoneinträge aus lexikon_np.pl⟩≡

```
% (definites) Personalpronomen
```

```
pron([], [1, -, sg, nom]) --> [ich].
```

```
pron([], [1, -, sg, gen]) --> [meiner].
```

```
pron([], [1, -, sg, dat]) --> [mir].
```

```
pron([], [1, -, sg, akk]) --> [mich].
```

```
pron([], [3, mask, sg, nom]) --> [er].
```

```
pron([], [3, mask, sg, gen]) --> [seiner].
```

```
pron([], [3, mask, sg, dat]) --> [ihm].
```

```
pron([], [3, mask, sg, akk]) --> [ihn].
```

```
% Eigennamen
```

```
pn([fem], [nom]) --> ['Maria'].
```

```
pn([fem], [gen]) --> ['Marias'].
```

```
pn([fem], [dat]) --> ['Maria'].
```

```
pn([fem], [akk]) --> ['Maria'].
```

...

<Lexikoneinträge aus lexikon_np.pl>≡

% Zaehlbare Nomen (count nouns):

cn([mask],[sg,nom]) --> ['Mensch'].

cn([mask],[sg,gen]) --> ['Menschen'].

cn([mask],[sg,dat]) --> ['Menschen'].

cn([mask],[sg,akk]) --> ['Menschen'].

...

% Massennomen

mn([mask],[sg,nom]) --> ['Wein'].

mn([mask],[sg,gen]) --> ['Weins'].

mn([mask],[sg,dat]) --> ['Wein'].

mn([mask],[sg,akk]) --> ['Wein'].

...

% Nominalisierte Verben nv([Kompl,Gen],[Num,Kas])

nv([[auf,akk],fem],[sg,nom]) --> ['Hoffnung'].

...

nv([[dass],fem],[sg,nom]) --> ['Behauptung'].

...

Adjektivphrasen

- ▶ Adjektiv mit Komplementen,
- ▶ nur im Positiv
- ▶ keine Modifikation durch Gradausdrücke/Adverbien
- ▶ nur zur Verwendung als Nomenattribut
- ▶ keine Vergleichskonstruktionen (so A er N wie NP)

Kategorie	Konstruktion	Beispiel
$AP^{Komp, Dekl, Gen, Num, Kas}$	$= (P_{Kas'} NP_{Pers', Gen', Num'}^{Kas'})$ $A_{[P, Kas']}^{Dekl, Gen, Num, Kas}$	auf sie
	$ (P_{Kas'} NP_{Pers', Gen', Num'}^{Kas'})$ $(NP_{Pers, Gen2, Num2}^{Kas2})$	neugieriges von uns
	$A_{[P, Kas', Kas2]}^{Dekl, Gen, Num, Kas}$	den Kindern empfohlenes

Implementierung als DCG

Adjektivphrasen (positiv) aus Adjektiv mit optionalem Komplement:

$\langle npgram.pl \rangle + \equiv$

```
% Adjektivphrasen: ap([], [Komp, Dekl, Gen, Num, Kas])

ap([], [pos, Dekl, Gen, Num, Kas]) -->
  ( [] % Komplemente stets optional
  ; { Komp1=[Pr, KasP], praeposition(Pr, KasP) },
    [Pr], np([_, -, _], [KasP])
  ; { Komp1=[Pr, KasP, KasA0], praeposition(Pr, KasP) },
    [Pr], np([_Pers, _Gen2, _Num2], [KasP]),
    (np([_PersA0, _GenA0, _NumA0], [KasA0]) ; [])
  ), % Linksrekursion durch Stellung vermieden!
  ( [] ; [sehr] ),
  a([Komp1], [Dekl, Gen, Num, Kas]).
```

Ausgeschlossen werden Adjektivphrasen, deren A-Komplemente mit A beginnen können: *ein (alten Menschen empfohlener):AP Wein.*

